
Graphics
Drawing Graphs with MetaPost
John D. Hobby
Abstract
This paper describes a graph-drawing package that
has been implemented as an extension to the MetaPost graphics language. MetaPost has a powerful
macro facility for implementing such extensions.
There are also some new language features that
support the graph macros. Existing features for
generating and manipulating pictures allow the user
to do things that would be diﬃcult to achieve in a
stand-alone graph package.
1

Introduction

MetaPost is a batch-oriented graphics language
based on Knuth’s METAFONT, but with PostScript
output and numerous features for integrating text
and graphics. The author has tried to make this
paper as independent as possible of the user’s manual [6], but fully appreciating all the material requires some knowledge of the MetaPost language.
We concentrate on the mechanics of producing
particular kinds of graphs because the question of
what type of graph is best in a given situation
is covered elsewhere; e.g., Cleveland [2, 4, 3] and
Tufte [11]. The goal is to provide at least the power
of UNIX grap [1], but within the MetaPost language.
Hence the package is implemented using MetaPost’s
powerful macro facility.
The graph macros provide the following functionality:
1. Automatic scaling
2. Automatic generation and labeling of tick
marks or grid lines
3. Multiple coordinate systems
4. Linear and logarithmic scales
5. Separate data ﬁles
6. Ability to handle numbers outside the usual
range
7. Arbitrary plotting symbols
8. Drawing, ﬁlling, and labeling commands for
graphs
In addition to these items, the user also has access
to all the features described in the MetaPost user’s
manual [6]. These include access to almost all the
features of PostScript, ability to use and manipulate
typeset text, ability to solve linear equations, and

data types for points, curves, pictures, and coordinate transformations.
Section 2 describes the graph macros from a
user’s perspective and presents several examples.
Sections 3 and 4 discuss auxiliary packages for manipulating and typesetting numbers and Section 5
gives some concluding remarks. Appendix A summarizes the graph-drawing macros, and Appendix B
describes some recent additions to the MetaPost
language that have not been presented elsewhere.
2

Using the Graph Macros

A MetaPost input ﬁle that uses the graph macros
should begin with
input graph
This reads a macro ﬁle graph.mp and deﬁnes the
graph-drawing commands explained below. The rest
of the ﬁle should be one or more instances of
beginfig(ﬁgure number);
graphics commands endfig;
followed by end.
The following graphics commands suﬃce to
generate the graph in Figure 1 from the data ﬁle
agepop91.d:
draw begingraph(3in,2in);
gdraw "agepop91.d";
endgraph;
(Each line of agepop91.d gives an age followed the
estimated number of Americans of that age in 1991
[10].)
2.1

Basic Graph-Drawing Commands

All graphs should begin with
begingraph(width,height);
and end with endgraph. This is syntactically a
picture expression, so it should be preceded by
draw and followed by a semicolon as in the example.1
The width and height give the dimensions of the
graph itself without the axis labels.
The command
gdraw expression option list
draws a graph line. If the expression is of type
string, it names a data ﬁle; otherwise it is a path
that gives the function to draw. The option list is
zero or more drawing options
withpen pen expression
| withcolor color expression
| dashed picture expression
1 See the user’s manual [6] for explanations of draw
commands and syntactic elements like picture expression.

Figure 1: A graph of the 1991 age distribution in the United States
that give the line width, color, or dash pattern as
explained in the User’s Manual [6].
In addition to the standard drawing options,
the option list in a gdraw statement can contain
plot picture expression
The picture expression gives a plotting symbol
to be drawn at each path knot. The plot option
suppresses line drawing so that2
gdraw "agepop91.d" plot btex • etex
generates only bullets as shown in Figure 2. (Following the plot option with a withpen option would
cause the line to reappear superimposed on the
plotting symbols.)
The glabel and gdotlabel commands add
labels to a graph. The syntax for glabel is
glabel.label suﬃx
(string or picture expression, location)
option list
where location identiﬁes the location being labeled
and label suﬃx tells how the label is oﬀset relative
to that location. The gdotlabel command is
identical, except it marks the location with a dot.
A label suﬃx is as in plain MetaPost: empty
centers the label on the location; lft, rt, top, bot
oﬀset the label horizontally or vertically; and ulft,
urt, llft, lrt give diagonal oﬀsets. The location
can be a pair of graph coordinates, a knot number
on the last gdraw path, or the special location OUT.
Thus
gdotlabel.top(btex (50, 0) etex, 50,0)
2 Troﬀ users should replace btex $\bullet$ etex with
btex \(bu etex.

would put a dot at graph coordinates (50,0) and
place the typeset text “(50, 0)” above it. Alternatively,
glabel.ulft("Knot3", 3)
typesets the string "Knot3" and places it above and
to the left of Knot 3 of the last gdraw path. (The
knot number 3 is the path’s “time” parameter [6,
Section 8.2].)
The location OUT places a label relative to
the whole graph. For example, replacing “gdraw
"agepop91.d"” with
glabel.lft(btex \vbox{\hbox{Population}
\hbox{in millions}} etex, OUT);
glabel.bot(btex Age in years etex, OUT)
gdraw "agepopm.d";
in the input for Figure 1 generates Figure 3. This
improves the graph by adding axis labels and using a
new data ﬁle agepopm.d where the populations have
been divided by one million to avoid large numbers.
We shall see later that simple transformations such
as this can be achieved without generating new data
ﬁles.
All ﬂavors of TEX can handle multi-line labels
via the \hbox within \vbox arrangement used above,
but LATEX users will ﬁnd it more natural to use the
tabular environment [9]. Troﬀ users can use noﬁll
mode:
btex .nf Population in millions etex
2.2

Coordinate Systems

The graph macros automatically shift and rescale
coordinates from data ﬁles, gdraw paths, and
glabel locations to ﬁt the graph. Whether the
range of y coordinates is 0.64 to 4.6 or 640,000 to
4,600,000, they get scaled to ﬁll about 88% of the

Figure 2: The 1991 age distribution plotted with bullets

Figure 3: An improved version of the 1991 age distribution graph
height speciﬁed in the begingraph statement. Of
course line widths, labels, and plotting symbols are
not rescaled.
The setrange command controls the shifting
and rescaling process by specifying the minimum
and maximum graph coordinates:
setrange(coordinates, coordinates)
where
coordinates → pair expression
| numeric or string expression,
numeric or string expression
The ﬁrst coordinates give (xmin , ymin ) and the
second give (xmax , ymax ). The lines x = xmin ,
x = xmax , y = ymin , and y = ymax deﬁne the
rectangular frame around the graph in Figures 1–
3. For example, an adding a statement
setrange(origin, whatever, whatever)

to the input for Figure 3 yields Figure 4. The
ﬁrst coordinates are given by the predeﬁned pair
constant origin, and the other coordinates are
left unspeciﬁed. Any unknown value would work
as well, but whatever is the standard MetaPost
representation for an anonymous unknown value.
Notice that the syntax for setrange allows
coordinate values to be given as strings. Many
commands in the graph package allow this option.
It is provided because the MetaPost language uses
ﬁxed point numbers that must be less than 32768.
This limitation is not as serious as it sounds because
good graph design dictates that coordinate values
should be “of reasonable magnitude” [2, 11]. If you
really want x and y to range from 0 to 1,000,000,
setrange(origin, "1e6", "1e6")

draw begingraph(3in,2in);
glabel.lft(btex \vbox{\hbox{Population} \hbox{in millions}} etex, OUT);
glabel.bot(btex Age in years etex, OUT);
setrange(origin, whatever,whatever);
gdraw "agepopm.d";
endgraph;

Figure 4: The 1991 age distribution graph and the input that creates it.
does the job. Any ﬁxed or ﬂoating point representation is acceptable as long as the exponent is
introduced by the letter “e”.
Coordinate systems need not be linear. The
setcoords command allows either or both axes to
have logarithmic spacing:
coordinate setting →
setcoords(coordinate type, coordinate type)
coordinate type →
log | linear | -log | -linear
A negative coordinate type makes x (or y) run
backwards so it is largest on the left side (or bottom)
of the graph.
Figure 5 graphs execution times for two matrix
multiplication algorithms using
setcoords(log,log)
to specify logarithmic spacing on both axes. The
data ﬁle matmul.d gives timings for both algorithms:
20
30
40
60
80
120
160
240
320
480

.007861
.022051
.050391
.15922
.4031
1.53
3.915
18.55
78.28
279.24

standard MM: size, seconds

20
30

.006611
.020820

Strassen: size, seconds

40
60
80
120
160
240
320
480

.049219
.163281
.3975
1.3125
3.04
9.95
22.17
72.60

A blank line in a data ﬁle ends a data set. Subsequent gdraw commands access additional data sets
by just naming the same data ﬁle again. Since each
line gives one x coordinate and one y coordinate,
commentary material after the second data ﬁeld on
a line is ignored.
Placing a setcoords command between two
gdraw commands graphs two functions in diﬀerent
coordinate systems as shown in Figure 6. Whenever
you give a setcoords command, the interpreter
examines what has been drawn, selects appropriate
x and y ranges, and scales everything to ﬁt. Everything drawn afterward is in a new coordinate system
that need not have anything in common with the
old coordinates unless setrange commands enforce
similar coordinate ranges. For instance, the two
setrange commands force both coordinate systems
to have x ranging from 80 to 90 and y starting at 0.
When you use multiple coordinate systems, you
have to specify where the axis labels go. The default
is to put tick marks on the bottom and the left
side of the frame using the coordinate system in

draw begingraph(2.3in,2in);
setcoords(log,log);
glabel.lft(btex Seconds etex,OUT);
glabel.bot(btex Matrix size etex,
OUT);
gdraw "matmul.d" dashed evenly;
glabel.ulft(btex Standard etex,8);
gdraw "matmul.d";
glabel.lrt(btex Strassen etex,7);
endgraph;

Figure 5: Timings for two matrix multiplication algorithms with the corresponding MetaPost input.

draw begingraph(6.5cm,4.5cm);
setrange(80,0, 90,whatever);
glabel.bot(btex Year etex, OUT);
glabel.lft(btex \vbox{\hbox{Emissions in} \hbox{thousands of}
\hbox{metric tons} \hbox{(heavy line)}}etex, OUT);
gdraw "lead.d" withpen pencircle scaled 1.5pt;
autogrid(,otick.lft);
setcoords(linear,linear);
setrange(80,0, 90,whatever);
glabel.rt(btex \vbox{\hbox{Micrograms} \hbox{per cubic}
\hbox{meter of air} \hbox{(thin line)}}etex, OUT);
gdraw "lead.d";
autogrid(otick.bot,otick.rt);
endgraph;

Figure 6: Annual lead emissions and average level at atmospheric monitoring stations in the United
States. The MetaPost input is shown above the graph.

eﬀect when the endgraph command is interpreted.
Figure 6 uses the
autogrid(,otick.lft)
to label the left side of the graph with the y coordinates in eﬀect before the setcoords command.
This suppresses the default axis labels, so another
autogrid command is needed to label the bottom
and right sides of the graph using the new coordinate
system. The general syntax is
autogrid(axis label command,
axis label command) option list
where
axis label command →
empty | grid or tick label suﬃx
grid or tick → grid | itick | otick
The label suﬃx should be lft, rt, top, or bot.
The ﬁrst argument to autogrid tells how to
label the x axis and the second argument does
the same for y. An empty argument suppresses
labeling for that axis. Otherwise, the label suﬃx
tells which side of the graph gets the numeric label.
Be careful to use bot or top for the x axis and lft
or rt for the y axis. Use otick for outward tick
marks, itick for inward tick marks, and grid for
grid lines. The option list tells how to draw the
tick marks or grid lines. Grid lines tend to be a
little overpowering, so it is a good idea to give a
withcolor option to make them light gray so they
do not make the graph too busy.
2.3

Explicit Grids and Framing

In case autogrid is not ﬂexible enough, axis label
commands generate grid lines or tick marks one at
a time. The syntax is
grid or tick.label suﬃx
(label format, numeric or string expression)
option list
where grid or tick and label suﬃx are as in
autogrid, and label format is either a format
string like "%g" or a picture containing the typeset
numeric label.
The axis label commands use a macro
format(format string,
numeric or string expression)
to typeset numeric labels. Full details appear in Section 4, but when the format string is "%g", it uses
decimal notation unless the number is large enough
or small enough to require scientiﬁc notation.
The example in Figure 7 invokes
format("
explicitly so that grid lines can be placed at transformed coordinates. It deﬁnes the transformation

newy(y) = y/75 + ln y and shows that this function
increases almost linearly.3 This is a little like using
logarithmic y-coordinates, except that y is mapped
to y/75 + ln y instead of just ln y.
Figure 7 uses the command
frame.label suﬃx option list
to draw a special frame around the graph. In
this case the label suﬃx is llft to draw just
the bottom and left sides of the frame. Suﬃxes
lrt, ulft, and urt draw other combinations of two
sides; suﬃxes lft, rt, top, bot draw one side, and
empty draws the whole frame. For example
frame dashed evenly
draws all four sides with dashed lines. The default
four-sided frame is drawn only when there is no
explicit frame command.
To label an axis as autogrid does but with the
labels transformed somehow, use
auto.x or auto.y
for positioning tick marks or grid lines. These
macros produce comma-separated lists for use in
for loops. Any x or y values in these lists that
cannot be represented accurately within MetaPost’s
ﬁxed-point number system are given as strings. A
standard macro package that is loaded via
input sarith
deﬁnes arithmetic operators that work on numbers
or strings. Binary operators Sadd, Ssub, Smul, and
Sdiv do addition, subtraction, multiplication, and
division.
One possible application is rescaling data. Figure 4 used a special data ﬁle agepopm.d that had
y values divided by one million. This could be
avoided by replacing “gdraw "agepopm.d"” by
gdraw "agepop91.d";
for u=auto.y:
otick.lft(format(" endfor
autogrid(otick.bot,)
2.4

Processing Data Files

The most general tool for processing data ﬁles is the
gdata command:
gdata(string expression, variable,
commands)
It takes a ﬁle name, a variable v, and a list of
commands to be executed for each line of the data
ﬁle. The commands are executed with i set to the
input line number and strings v1, v2, v3, . . . set to
the input ﬁelds on the current line. A null string
marks the end of the v array.
3 The manual [6] explains how vardef deﬁnes functions
and mlog computes logarithms.

vardef newy(expr y) = (256/75)*y + mlog y enddef;
draw begingraph(3in,2in);
glabel.lft(btex \vbox{\hbox{Population} \hbox{in millions}} etex, OUT);
gdraw "ttimepop.d";
for y=5,10,20,50,100,150,200,250:
grid.lft(format("%g",y), newy(y)) withcolor .85white;
endfor
autogrid(grid.bot,) withcolor .85white;
frame.llft;
endgraph;
250

10

1800

Figure 7: Population of the United States in millions versus time with the population re-expressed as
p/75 + ln p. The MetaPost input shown above the graph assumes a data ﬁle ttimepop.d that gives (year,
p/75 + ln p) pairs.
Using a glabel command inside of gdata generates a scatter plot as shown in Figure 8. The data
ﬁle countries.d begins
20.910 75.7 US
1.831 66.7 Alg
where the last ﬁeld in each line gives the label to
be plotted. Setting defaultfont in the ﬁrst line of
input selects a small font for these labels. Without
these labels, no gdata command would be needed.
Replacing the gdata command with
gdraw "countries.d" plot btex $\circ$etex
would change the abbreviated country names to
open circles.
Both gdraw and gdata ignore an optional initial
‘%’ on each input line, parse data ﬁelds separated by
white space, and stop if they encounter an input
line with no data ﬁelds. Leading percent signs make
graph data look like MetaPost comments so that
numeric data can be placed at the beginning of a
MetaPost input ﬁle.
It is often useful to construct one or more paths
when reading a data ﬁle with gdata. The augment
command is designed for this:
augment.path variable (coordinates)

If the path variable does not have a known value, it
becomes a path of length zero at the given coordinates; otherwise a line segment to the given coordinates is appended to the path. The coordinates
may be a pair expression or any combination of
strings and numerics as explained at the beginning
of Section 2.2.
If a ﬁle timepop.d gives t, p pairs, augment can
be used like this to graph newy(p) versus t:
path p;
gdata("timepop.d", s,
augment.p(s1, newy(scantokens s2)); );
gdraw p;
(MetaPost’s scantokens primitive interprets a
string as if it were the contents of an input ﬁle. This
ﬁnds the numeric value of data ﬁeld s2.)
Figure 9 shows how to use augment to read
multiple column data and make multiple paths.
Paths p2, p3, p4, p5 give cumulative totals for
columns 2 through 5 and pictures lab2 through lab5
give corresponding labels. The expression
image(unfill bbox lab[j]; draw lab[j])
executes the given drawing commands and returns
the resulting picture: “unfill bbox lab[j]” puts

defaultfont:="cmr7";
draw begingraph(3in,2in);
glabel.lft(btex \vbox{\hbox{Life}\hbox{expectancy}} etex, OUT);
glabel.bot(btex Per capita G.N.P. (thousands of dollars) etex, OUT);
setcoords(log,linear);
gdata("countries.d", s,
glabel(s3, s1, s2);
)
endgraph;

Figure 8: A scatter plot and the commands that generated it
down a white background and “draw lab[j]” puts
the label on the background. The gfill command
is just like gdraw, except it takes a cyclic path and
ﬁlls the interior with a solid color. The color is black
unless a withcolor clause speciﬁes another color.
See the manual [6] for explanations of for loops,
arrays, colors, and path construction operators like
--, cycle, and reverse.
3

Manipulating Big Numbers

MetaPost inherits a ﬁxed-point number system from
Knuth’s METAFONT [8]. Numbers are expressed
in multiples of 2−16 and they must have absolute
value less than 32768. Knuth chose this system
because it is perfectly adequate for font design, and
it is guaranteed to give identical results on all types
of computers. Fixed-point numbers are seldom a
problem in MetaPost because all computations are
based on coordinates that are limited by the size
of the paper on which the output is to be printed.
This does not hold for the input data in a graphdrawing application. Although graphs look best
when coordinate axes are labeled with numbers of
reasonable magnitude, the strict limits of ﬁxed-point
arithmetic would be inconvenient.

A simple way to handle large numbers is to
include the line
input sarith
and then use binary operators Sadd, Ssub, Smul, and
Sdiv in place of +, -, *, and /. These operators are
ineﬃcient but very ﬂexible. They accept numbers
or strings and return strings in exponential notation
with the exponent marked by “e”; e.g., "6.7e-11"
means 6.7 × 10−11 .
The unary operator4
Sabs string
ﬁnds a string that represents the absolute value.
Binary operators Sleq and Sneq perform numeric
comparisons on strings and return boolean results.
The operation
Scvnum string
ﬁnds the numeric value for a string if this can
be done without overﬂowing MetaPost’s ﬁxed-point
number system. If the string does not contain “e”, it
is much more eﬃcient to use the primitive operation
scantokens string
The above operators are based on a low-level
package that manipulates numbers in “Mlog form.”
4 The argument to a unary operator need not be parenthesized unless it is an expression involving binary operators.

draw begingraph(3in,2in);
glabel.lft(btex \vbox{\hbox{Quadrillions}\hbox{of BTU}} etex, OUT);
path p[];
numeric t;
gdata("energy.d", $,
t:=0; augment.p1($1,0);
for j=2 upto 5:
t:=t+scantokens $[j]; augment.p[j]($1,t);
endfor)
picture lab[];
lab2=btex coal etex; lab3=btex crude oil etex;
lab4=btex natural gas etex; lab5=btex hydroelectric etex;
for j=5 downto 2:
gfill p[j]--reverse p[j-1]--cycle withcolor .16j*white;
glabel.lft(image(unfill bbox lab[j]; draw lab[j]), .7+length p[j]);
endfor
endgraph;
hydroelectric

60

40
Quadrillions

20

coal
0
1900

1960

Figure 9: A graph of U.S. annual energy production and the commands that generated it
A number x in Mlog form represents
16

µ2

x

,

−24

where µ = −e2

interim Gpaths:=log;
gdraw Mreadpath("agepop91.d")
shifted (0,-6*Mten);

.

Any value between 1.61 × 10−28 and 3.88 × 1055 can
be represented this way. (There is a constant Mten
such that k ∗ Mten represents 10k for any integer k
in the interval [−29, 55].)
The main reason for mentioning Mlog form is
that it allows graph data to be manipulated as a
MetaPost path. The function
Mreadpath(ﬁle name)
reads a data ﬁle and returns a path where all the
coordinates are in Mlog form. An internal variable
Gpaths determines whether gdraw and gfill expect
paths to be given in Mlog form. For example, this
graphs the data in agepop91.d with y coordinates
divided by one million:

4

Typesetting Numbers

The graph package needs to compute axis labels and
then typeset them. The macro
format(string expression,
numeric or string expression)
does this. You must ﬁrst input graph or input
format to load the macro ﬁle. The macro takes a
format string and a number to typeset and returns
a picture containing the typeset result. Thus
format("%g",2+2) yields

4

and
format("%3g","6.022e23") yields
A format string consists of

6.02×1023

• an optional initial string not containing a percent sign,
• a percent sign,
• an optional numeric precision p,
• one of the conversion letters e, f, g, G,
• an optional ﬁnal string β.
The initial and ﬁnal strings are typeset in the default
font (usually cmr10), and the typeset number is
placed between them. For the e and g formats, the
precision p is the number of signiﬁcant digits allowed
after rounding; for f and G, the number is rounded
to the nearest multiple of 10−p . If the precision is
not speciﬁed, the default is p = 3. The e format
always uses scientiﬁc notation and the f format uses
ordinary decimal notation but reverts to scientiﬁc
notation if the number is at least 10000. The g and
G formats also revert to scientiﬁc notation for nonzero numbers of magnitude less than 0.001.
The format macro needs a set of templates
to determine what font to use, how to position
the exponent, etc. The templates are normally
initialized automatically, but it is possible to set
them explicitly by passing ﬁve picture expressions to
init_numbers. For instance, the default deﬁnition
for TEX users is
init_numbers(btex$-$etex, btex$1$etex,
btex${\times}10$etex,
btex${}^-$etex, btex${}^2$etex)
The ﬁrst argument tells how to typeset a leading
minus sign; the second argument is an example of
a 1-digit mantissa; third comes whatever to put
after the mantissa in scientiﬁc notation; next come
a leading minus sign for the exponent and a sample
1-digit exponent.
Picture variable Fe_plus gives a leading plus
sign for positive numbers, and Fe_base gives whatever should precede the exponent when typesetting a power of ten. Calling init_numbers initializes Fe_plus to an empty picture and constructs
Fe_base from its second and third arguments.
5

Conclusion

The graph package makes it convenient to generate
graphs from within the MetaPost language. The
primary beneﬁts are the power of the MetaPost
language and its ability to interact with TEX or troﬀ
for typesetting labels. Typeset labels can be stored
in picture variables and manipulated in various ways
such as measuring the bounding box and providing
a white background.
We have seen how to generate shaded regions
and control line width, color, and styles of dashed
lines. Numerous other variations are possible. The

full MetaPost language [6] provides many other
potentially useful features. It also has enough
computing power to be useful for generating and
processing data.
A

Summary of the Graph Package

In the following descriptions, italic letters such as
w and h denote expression parameters and words
in angle brackets denote other syntactic elements.
Unless speciﬁed otherwise, expression parameters
can be either numerics or strings. An option
list is a list of drawing options such as withcolor
.5white or dashed evenly; a label suﬃx is one
of lft, rt, top, bot, ulft, urt, llft, lrt.
A.1

Graph Administration

begingraph(w,h) Begin a new graph with the
frame width and height given by numeric parameters w and h.
endgraph End a graph and return the resulting
picture.
setcoords(tx, ty ) Set up a new coordinate system
as speciﬁed by numeric ﬂags tx , ty . Flag values
are ±linear and ±log.
setrange(coordinates, coordinates) Set the
lower and upper limits for the current coordinate system. Each coordinates can be a
single pair expression or two numeric or string
expressions.
A.2

Drawing and Labeling

All of the drawing and labeling commands can be
followed by an option list. In addition to the usual
MetaPost drawing options, the list can contain a
plot picture clause to plot a speciﬁed picture at
each data point.
The drawing and labeling commands are
closely related to a set of similarly named commands in plain MetaPost. The gdrawarrow and
gdrawdblarrow commands are included to maintain
this relationship.
gdotlabel.label suﬃx(p, location) This is like
glabel except it also puts a dot at the location
being labeled.
gdraw p Draw path p, or if p is a string, read coordinate pairs from ﬁle p and draw a polygonal
line through them.
gdrawarrow p This is like dgraw p except it adds
an arrowhead at the end of the path.
gdrawdblarrow p This is like dgraw p except it
adds an arrowhead at each end of the path.
gfill p Fill cyclic path p or read coordinates from
the ﬁle named by string p and ﬁll the resulting
polygonal outline.

glabel.label suﬃx(p, location) If p is not a
picture, it should be a string. Typeset it
using defaultfont, then place it near the
given location and oﬀset as speciﬁed by the
label suﬃx. The location can be x and y
coordinates, a pair giving x and y, a numeric
value giving a time on the last path drawn, or
OUT to label the outside of the graph.
A.3

Grids, Tick Marks, and Framing

auto.x or y Generate default x or y coordinates
for tick marks.
autogrid(axis label command,
axis label command)
Draw default axis labels using the speciﬁed
commands for the x and y axes. An axis label
command may be empty or it may be itick,
otick, or grid followed by a label suﬃx.
frame.label suﬃx option list Draw a frame
around the graph, or draw the part of the frame
speciﬁed by the label suﬃx.
grid.label suﬃx(f ,z) Draw a grid line across the
graph from the side speciﬁed by the label
suﬃx, and label it there using format string f
and coordinate value z. If f is a picture, it gives
the label.
itick.label suﬃx(f ,z) This is like grid except it
draws an inward tick mark.
otick.label suﬃx(f ,z) This is like grid except it
draws an outward tick mark.
A.4

Miscellaneous Commands

augment.variable(coordinates) Append coordinates to the path stored in variable.
format(f , x) Typeset x according to format string f
and return the resulting picture.
gdata(f , variable, commands) Read the ﬁle
named by string f and execute commands for
each input line using the variable as an array
to store data ﬁelds.
init numbers(s, m, x, t, e) Provide ﬁve pictures
as templates for future format operations: s
is a leading minus; m is a sample mantissa; x
follows the mantissa; t is a leading minus for
the exponent e.
Mreadpath(f ) Read a path for the data ﬁle named
by string f and return it in “Mlog form”.
A.5

Arithmetic on Numeric Strings

It is necessary to input sarith before using the
following macros:
Sabs x Compute |x| and return a numeric string.

x Sadd y Compute x + y and return a numeric
string.
Scvnum x Return the numeric value for string x.
x Sdiv y Compute x/y and return a numeric
string.
x Sleq y Return the boolean result of the comparison x ≤ y.
x Smul y Compute x ∗ y and return a numeric
string.
x Sneq y Return the boolean result of the comparison x = y.
x Ssub y Compute x − y and return a numeric
string.
A.6

Internal Variables and Constants

Autoform Format string used by autogrid. Default: "%g".
Fe base What precedes the exponent when typesetting a power of ten.
Fe plus Picture of the leading plus sign for positive
exponents.
Gmarks Minimum number of tick marks per axis for
auto and autogrid. Default: 4.
Gminlog Minimum largest/smallest ratio for logarithmic spacing with auto and autogrid. Default: 3.0.
Gpaths Code for coordinates used in gdraw and
gfill paths: linear for standard form, log for
“Mlog form”.
Mten The “Mlog form” for 10.0
B

New Language Features

The graph.mp macros and the arithmetic routines
in marith.mp and sarith.mp use various language
features that were introduced in Version 0.60 of the
MetaPost language. We summarize these features
here because they are not covered in existing documentation [6, 5]. Also new is the built-in macro
image(drawing commands)
that was used in Section 2.4 to ﬁnd the picture
produced by a sequence of drawing commands.
B.1

Reading and Writing Files

A new operator
readfrom ﬁle name
returns a string giving the next line of input from
the named ﬁle. The ﬁle name can be any primary
expression of type string. If the ﬁle has ended or
cannot be read, the result is a string consisting of a
single null character. The preloaded plain macro
package introduces the name EOF for this string.

After readfrom has returned EOF, additional reads
from the same ﬁle cause the ﬁle to be reread from
the start.
The opposite of readfrom is the command
write string expression toﬁle name
This writes a line of text to the speciﬁed output ﬁle,
opening the ﬁle ﬁrst if necessary. All such ﬁles are
closed automatically when the program terminates.
They can also be closed explicitly by using EOF as
the string expression. The only way to tell if a
write command has succeeded is to close the ﬁle
and use readfrom to look at it.
B.2

Extracting Information from Pictures

MetaPost pictures are composed of stroked lines,
ﬁlled outlines, pieces of typeset text, clipping paths,
and setbounds paths. (A setbounds path gives an
artiﬁcial bounding box as is needed for TEX output.)
A picture can have many components of each type.
They can be accessed via an iteration of the form
for symbolic token within picture expression:
loop text endfor
The loop text can be anything that is balanced
with respect to for and endfor. The symbolic
token is a loop variable that scans the components
of the picture in the order in which they were
drawn. The component for a clipping or setbounds
path includes everything the path applies to. Thus
if a single clipping or setbounds path applies to
everything in the picture expression, the whole
picture could be thought of as one big component.
In order to make the contents of such a picture
accessible, the for. . . within iteration ignores the
enclosing clipping or setbounds path in this case.
Once the for. . . within iteration has found a
picture component, there are numerous operators
for identifying it and extracting relevant information. The operator
stroked primary expression
tests whether the expression is a known picture
whose ﬁrst component is a stroked line. Similarly,
the filled and textual operators return true if
the ﬁrst component is a ﬁlled outline or a piece of
typeset text. The clipped and bounded operators
test whether the argument is a known picture that
starts with a clipping path or a setbounds path.
This is true if the ﬁrst component is clipped or
bounded or if the entire picture is enclosed in a
clipping or setbounds path.
There are also numerous part extraction operators that test the ﬁrst component of a picture. If p is
a picture and stroked p is true, pathpart p is the

path describing the line that got stroked, penpart
p is the pen that was used, dashpart p is the dash
pattern, and the color is
(redpart p, greenpart p, bluepart p)
If the line is not dashed, dashpart p returns an
empty picture.
The same part extraction operators work when
filled p is true, except that dashpart p is
not meaningful in that case. For text components, textual p is true, textpart p gives the
text that got typeset, fontpart p gives the font
that was used, and xpart p, ypart p, xxpart p,
xypart p, yxpart p, yypart p tell how the text
has been shifted, rotated, and scaled. The redpart,
greenpart, and bluepart operators also work for
text components.
When clipped p or bounded p is true,
pathpart p gives the clipping or setbounds path
and the other part extraction operators are not
meaningful. Such non-meaningful part extractions
do not generate errors — they return null values instead: the trivial path (0,0) for pathpart, nullpen
for penpart, an empty picture for dashpart, zero for
redpart, greenpart, bluepart, and the null string
for textpart or fontpart.
One ﬁnal operator for extracting information
from a picture is
length picture primary
This returns the number of components that a
for. . . within iteration would ﬁnd.
B.3

Other New Features

The marith.mp and sarith.mp packages use numbers of magnitude 4096 or more. Since such numbers
can cause overﬂow problems in MetaPost’s linear
equation solving and path ﬁtting algorithms, they
are normally allowed only as intermediate results.
This limitation is removed when the internal variable warningcheck is zero. In earlier versions of
MetaPost, the limitation could be removed for variables but explicit constants were always restricted
to be less than 4096.
For completeness, we also mention one other
new feature of MetaPost Version 0.60. When TEX
material is included in a picture via the btex. . . etex
feature, the thickness of horizontal and vertical
rules gets rounded to exactly the right number of
pixels; i.e., interpreting MetaPost output according
to the PostScript scan conversion rules [7] makes the
pixel width equal to the ceiling of the unrounded
width. In fact, a similar relationship holds for all
line widths. The generated PostScript sets line

widths by ﬁrst transforming to device coordinates
and rounding appropriately.

